
ABSTRACT 
This article provides a comprehensive guide to confirming Best Overall Response (BOR) in oncology 
clinical trials, using RECIST 1.1 criteria. It outlines the process for confirming BOR, emphasizing the 
importance of confirmation for Complete Response (CR) and Partial Response (PR) in non-randomized 
trials with disease response as the primary endpoint. The article is divided into three parts: a summary of 
RECIST 1.1 related to BOR confirmation, practical exercises, and SAS programming examples for BOR 
confirmation. This resource aims to enhance understanding and application of BOR confirmation in 
clinical research. 

INTRODUCTION 
Confirmation of response is required for non-randomized trials with response primary endpoint. This 
article introduces the confirmation of BOR based on RECIST1.1. 

The article consists of three parts. The first part describes the confirmation of BOR in RECIST 1.1 and 
provides additional related content. The second part includes instructions and demonstrations for BOR 
confirmation exercises. The third part shares SAS programs for BOR confirmation. 

It is hoped that the article and its accompanying exercises will help readers better understand and master 
the confirmation of BOR. After completing the exercises, everyone should be able to determine 
BOR_temp within 3 seconds. 

1. SUMMARY OF RECIST 1.1  
RECIST 1.1, Response Evaluation Criteria in Solid Tumors (Version 1.1), is the standard for assessing 
response in solid tumors. The specific process of evaluating solid tumor involves selecting a certain 
number of target lesions and non-target lesions at baseline and calculating the sum of the diameters of all 
target lesions. At subsequent time points after the baseline, the sum of the diameters of the target lesions 
is also calculated and compared with the baseline to assess the response of the target lesions at that 
time point; simultaneously, researchers qualitatively assess the response of the non-target lesions, 
observe and record whether new lesions appear.   

Combining the response of target and non-target lesions and the appearance of new lesions, researchers 
evaluate the overall response of the subject at that time point (Time point response). By considering the 
responses at all time points, the BOR is determined. 

1.1. TIME POINT RESPONSE 
After selecting target and non-target lesions at baseline, the response of the tumor at each subsequent 
time point is assessed based on the observations of the target lesions, non-target lesions, and any new 
lesions. The assessment rules are as follows: 

Table 1. Time point response: patients with target (+/- non-target) disease 
Time point response is usually provided by on-site researchers and independent central reviewers, and 
does not need to be separately derived by the sponsor. This information is typically processed in the 
SDTM RS domain. 

1.2. BEST OVERALL RESPONSE: ALL TIME POINTS 
Once the time point responses for a subject are obtained, the BOR can be determined. Intuitively, the 
BOR is the assessment with the best response among all time points. The priority of response 
comparison is as follows: CR > PR > SD > PD > NE. 

For example, if the time point responses after baseline are PR-SD-SD-PD, then PR would be considered 
the BOR if confirmation is not required.  

In some trials, the BOR cannot be determined by simply comparing time point responses. RECIST 1.1 
categorizes BOR determination into two types. (RECIST 1.1, Section 4.4.4) 

• Confirmation of complete or partial response IS NOT required.  

• Confirmation of complete or partial response IS required.  

If confirmation of response is not required, the best time point response after baseline can be directly 
selected as the BOR. If confirmation is required, the time point response cannot be directly used to 
determine the BOR.  

Generally, on-site researchers only provide the time point responses for subjects, not the BOR. The BOR 
needs to be derived separately. Independent central reviewers might provide the BOR in addition to the 
time point responses. In such cases, separate derivation is not needed. Typically, the BOR is processed 
in the ADaM ADRS domain. 

1.3. CONFIRMATION 
Section 4.6.1 of RECIST 1.1 discusses the need for confirmation of tumor responses: In non-randomized 
trials where disease response (CR+PR) is the primary endpoint, it is important to confirm CR and PR to 
ensure that the response is not due to measurement error. Besides CR and PR, the original text also 
mentions that SD needs confirmation. 

In some studies, based on the specifics of the trial, PD may also need confirmation. This should be 
specified in the Protocol and SAP requirements. In such cases, the response assessment is usually noted 
as "Modified RECIST 1.1". According to the original RECIST 1.1 guidelines, PD does not require 
confirmation. If PD occurs, the scheduled response assessments may be paused, and new treatments 
may be adopted for the subject based on the actual situation. 

1.4. HOW TO CONFIRM? 

1.4.1. Confirmation of CR and PR 
Confirmation of CR and PR requires subsequent time point responses to meet specific standards, which 
include both response and timing criteria. (RECIST 1.1, Section 4.4.2)  

For response requirements: 

• CR can only be confirmed by a subsequent CR.  

• PR can be confirmed by a subsequent CR or PR. 

For timing requirements: 

• The interval between the initial and subsequent time points usually needs to be at least 4 weeks. 

1.4.2. Confirmation of SD 
Although RECIST 1.1 does not clearly describe the confirmation of SD, it also needs to meet response 
and timing criteria: (RECIST 1.1, Section 4.6.1) 

• For response, the response must be SD or better (SD, PR, or CR). 

• For timing, the response must meet a minimum duration from the start of the study before it can be 
confirmed as SD, typically no less than 6-8 weeks, though specific requirements depend on the trial 
analysis. 

1.4.3. Confirmation of PD 
RECIST 1.1 does not require confirmation for PD; it can be understood that PD itself confirms PD. If a trial 
does require PD confirmation, the process is similar to that for confirming CR and PR, and this article 
does not provide additional details on this. 

1.4.4. Case of NE 
If a response cannot be confirmed as CR, PR, SD, or PD, then the corresponding BOR is classified as 
NE. 

1.4.5. Selection of Confirmed Results 
CR and PR require confirmation by a subsequent time point response, while SD can be confirmed by its 
own criteria. This may result in two confirmed responses. The process of confirming BOR involves 
identifying the best response. If a time point response can be confirmed as both PR and SD, the better 
response (PR) is selected as the BOR. 

1.5. CLASSIFICATION OF BOR CONFIRMATION 
In RECIST 1.1, BOR confirmation is divided into two types:  

• Confirmation of adjacent time point responses 

• Confirmation of non-adjacent time point responses 

1.5.1. Confirmation of Responses at Adjacent Time Points 
The original document provides a table to describe the confirmation of adjacent time point responses 
(RECIST 1.1, Section 4.4.3). This table is rather brief and has some limitations. It assumes that the 
interval between two adjacent time point responses is more than 4 weeks and does not cover all possible 
scenarios. 

 
Table 2. Best overall response when confirmation of CR and PR required. 
The table in the original RECIST 1.1 document is a rough reference. In actual projects, the SAP generally 
provides a more comprehensive description, and the actual programming needs to refer to the project's 
analysis specifications. 

Table 3. BOR per RECIST 1.1 
The above table describes the confirmation of adjacent time point responses. During this process, there 
may be multiple confirmed results, and BOR will select the best-confirmed response. 

For a subject, there will only be one BOR. The confirmation results of two adjacent time point responses 
can be called temporary BOR (BOR_temp). The final BOR is selected by combining all confirmed 
responses at all time points and choosing the best one. The current time point's BOR_temp is determined 
by comparing the confirmed responses of the current and previous time points and selecting the better 
one. 

For the first time point response after baseline, since there is no previous response to compare with, 
BOR_temp can only be confirmed from the first time point itself. This corresponds to the "No further 
evaluations" scenario in the table mentioned earlier. In other words, for the first time point response, 
BOR_temp can only be SD, PD, or NE, as there are no CR or PR responses from previous time points to 
confirm against. 

Next, I will explain in detail all the confirmation scenarios in the RECIST 1.1 original table to help readers 
better understand the confirmation of adjacent time point BOR. 

Scenario 1 
If the T1 time point response is CR, and the interval condition of 4 weeks is met, the T2 time point 
response of CR can confirm the T1 time point CR. In this case, the BOR at T2 is CR. 

Scenario 2 
If T1 is CR and T2 (after 4 weeks) is PR, this might indicate a data issue, requiring re-examination. 
Footnote a explains: 

• If CR was indeed achieved at T1, any lesion observed at a subsequent time point, even one meeting 
PR criteria relative to baseline, would make that lesion PD at that time point (as the lesion reappeared 
after CR). 

• The best response depends on meeting the minimum SD duration. 

• If subsequent scans show minor lesions suggesting PR at T1 rather than CR, the initial CR should be 
changed to PR, making the best response PR. 

Detailed explanation: 

• T1 may actually be PR. If T2 is PR, T2 PR confirms T1 PR, so BOR is PR. 

• T2 may actually be PD, and PD confirms itself. If T1 CR meets the minimum SD duration, CR 
confirms as SD; since CR cannot confirm subsequent PD, BOR is SD (SD > PD). 

• T2 may actually be PD, and PD confirms itself. If T1 CR does not meet the minimum SD duration, CR 
confirms as NE; since CR cannot confirm subsequent PD, BOR is PD (PD > NE). 

Scenario 3 
If T1 is CR and T2 (after 4 weeks) is SD, this is similar to Scenario 2 and might indicate a data issue 
requiring re-examination. The original text lacks precision. 

Detailed explanation: 

• T1 may actually be PR, and T2 is SD. If T1 or T2 meets the minimum SD duration, BOR is SD. 

• T1 may actually be PR, and T2 is SD. If neither T1 nor T2 meets the minimum SD duration, BOR is 
NE. 

• T2 may actually be PD, and PD confirms itself. If T1 CR does not meet the minimum SD duration, CR 
confirms as NE; since CR cannot be confirmed by subsequent PD, BOR is PD (PD > NE). 

Scenario 4 
If T1 is CR and T2 (after 4 weeks) is PD, PD confirms itself. If T1 does not meet the minimum SD 
duration, CR is confirmed as NE, so BOR is PD (PD > NE). If T1 meets the minimum SD duration, CR is 
confirmed as SD, so BOR is SD (SD > PD). 

Scenario 5 
If T1 is CR and T2 (after 4 weeks) is NE, if T1 does not meet the minimum SD duration, CR is confirmed 
as NE, so BOR is NE. If T1 meets the minimum SD duration, CR is confirmed as SD, so BOR is SD 
(SD > NE). 

Scenario 6 
If T1 is PR and T2 (after 4 weeks) is CR, CR can confirm PR. If T1 or T2 meets the minimum SD duration, 
it is confirmed as SD, so the final BOR is PR (PR > SD). If T1 or T2 does not meet the minimum SD 
duration, it is confirmed as NE, so the final BOR is PR (PR > NE). 

Scenario 7 
If T1 is PR and T2 (after 4 weeks) is PR, it is the same as Scenario 6. 

Scenario 8 
If T1 is PR and T2 (after 4 weeks) is SD, the original text lacks precision. If T1 or T2 meets the minimum 
SD duration, BOR is SD. If neither T1 nor T2 meets the minimum SD duration, BOR is NE. 

Scenario 9 
If T1 is PR and T2 (after 4 weeks) is PD, refer to the second explanation in Scenario 1. 

Scenario 10 
If T1 is PR and T2 (after 4 weeks) is NE, if T1 meets the minimum SD duration, it is confirmed as SD, so 
the final BOR is SD (SD > NE). If T1 does not meet the minimum SD duration, it is confirmed as NE, so 
the final BOR is NE. 

Scenario 11 
If T1 is NE and T2 (after 4 weeks) is NE, the BOR is clearly NE. 

1.5.2. Confirmation of Non-Adjacent Time Point Responses 
The description of non-adjacent time point response confirmation is limited in the original document, but 
Section 4.4.4 states: 

In trials requiring response confirmation, repeated "NE" assessments at different time points may 
complicate the determination of the best response. The trial analysis plan must address how to handle 
missing data/assessments when determining response and progression. For example, in most trials, it is 
reasonable to consider a patient with time point responses of PR-NE-PR as having a confirmed response. 

Generally, intermediate NE responses, regardless of the number, do not affect the confirmation of CR 
and PR. For instance, the following scenarios confirm CR and PR: 

• CR-NE-CR 

• CR-NE-NE-CR 

• PR-NE-PR/CR 

• PR-NE-NE-PR/CR 

The original document’s Appendix 3 FAQ mentions the confirmation of PR-SD-PR: 

• Q: A patient has a 32% decrease in sum cycle 2, a 28% decrease cycle 4 and a 33% decrease cycle 
6. Does confirmation of PR have to take place in sequential scans or is a case like this confirmed 
PR? 

• A: It is not infrequent that tumour shrinkage hovers around the 30% mark. In this case, most would 
consider PR to have been confirmed looking at this overall case. Had there been two or three non-PR 
observations between the two time point PR responses, the most conservative approach would be to 
consider this case SD. 

Although the original text mentions that PR-SD-PR can confirm PR, the final determination should follow 
the project's specific guidelines. 

Another type of non-adjacent time point response confirmation includes: 

• PR-PR-PR/CR 

• CR-CR-CR 

When there are two consecutive PRs, but the interval between the first and second PR does not exceed 
28 days, PR cannot be confirmed. However, if the third response is PR or CR, some projects may specify 
that the first PR can be confirmed by the third PR. 

The same logic applies to CR. The specific conditions should follow the study analysis requirements. 

2. EXERCISES FOR BOR CONFIRMATION 
I simulated a dozen examples for readers to practice confirming BOR. Below are a few examples for 
demonstration. For complete examples, refer to the attachment. 

The exercises are divided into two parts. The first part (ADRS_1) focuses on confirming BOR considering 
only adjacent time points. The second part (ADRS_2) involves confirming BOR considering both adjacent 
and non-adjacent time points. The answers are provided in the ADRS_1_AN and ADRS_2_AN sheets. If 
there is any doubt about the answers, you can refer to the content of BOR_PER_RECIST_1.1. 

Once you are very proficient at manually determining BOR, you can import the RawData RS into SAS 
and use a program to confirm BOR. Then, compare the program's output with your manual 
determinations. 

2.1. CONFIRMATION OF RESPONSES AT ADJACENT TIME POINTS 

 
Table 4. Cconfirmation of Responses at Adjacent Time Points 
In the top left corner of the ADRS_1 sheet, you will find relevant confirmation requirements. The two most 
important points are: 

• The minimum duration for SD is 35 days. 

• The interval between subsequent time points for CR and PR confirmation must be ≥ 28 days. 

Note that for the interval from C1D1, the exercise uses ADY, which is the value of T1-RFSDTC+1. 
Different projects may have different requirements; some may use T1-RFSTDTC or the latest TR 
assessment value at baseline as the subtrahend. The minimum interval between adjacent time points 
also depends on specific project requirements. 

Subject 001 

 
Table 5. Subject 001 
If the T1 time point after baseline is PR and ADY=38 exceeds the minimum time limit of SD, it can be 
confirmed that the BOR_temp is SD. Since there is no adjacent time point at T1, the BOR_temp is SD. 

If the time point of T2 is PR and ADY=85 exceeds the minimum time limit of SD, the BOR_temp can be 
confirmed as SD. T2-T1=47≥28, the PR at T2 can confirm the PR at T1, and the PR at T1 can be 
confirmed by BOR_temp. Finally BOR_temp better response PR. 

If the time point of T3 is SD and ADY=123 exceeds the minimum time limit of SD, the BOR_temp can be 
confirmed as SD. If the time point of T2 is PR and ADY=85 exceeds the minimum time limit of SD, the 
BOR_temp can be confirmed as SD. The T3 SD cannot confirm the PR at the T2 time, and the last 
BOR_temp is SD. 

USUBJID Time_Point ADY RSDY_lag DIFF PARAM PARAMCD RSSTRESC RSEVAL BOR_TEMP1 BOR
001 2 38  Overall Response OVRLRESP PR INVESTIGATOR SD SD
001 3 85 38 47 Overall Response OVRLRESP PR INVESTIGATOR PR PR
001 4 123 85 38 Overall Response OVRLRESP SD INVESTIGATOR SD PR
001 5 165 123 42 Overall Response OVRLRESP PD INVESTIGATOR SD PR

The time point of T4 is PD, and the BOR_temp is directly confirmed to be PD; If the time point of T3 is SD 
and ADY=123 exceeds the minimum time limit of SD, the BOR_temp can be confirmed as SD. 
Eventually, the BOR_temp is SD. 

After obtaining the BOR_temp values at each time point, the optimal response is the subject's BOR. For 
each subject, there will only be one record of the best overall effect. However, BOR may be derived 
differently by different companies. Our current practice is to keep a BOR record for each time point, which 
means the best overall reponse up to the current time point. 

Taking subject 001 as an example, for the time point T1, the BOR was SD; for T2, BOR is PR; for T3, the 
BOR is PR; for T4, the BOR is PR. 

Subject 012 

 
Table 6. Subject 012 
After the baseline, the T1 time point was CR, and ADY=30 did not exceed the minimum time limit of SD, 
so it could not be confirmed that the BOR_temp was SD and the BOR_temp was NE. Because there is no 
adjacent time point at T1, the BOR_temp is NE. 

The T2 time point after baseline is CR, and ADY=56 exceeds the minimum time limit of SD, which can 
confirm that the BOR_temp is SD. Since T2-T1=26<28, the CR at T2 could not confirm the CR at T1. 
Finally, the BOR_temp is SD. 

After baseline, the T3 time point is CR, and ADY=83 exceeds the minimum time limit of SD, which can 
confirm that the BOR_temp is SD. Since T3-T2=27<28, the CR at T3 could not confirm the CR at T2. 
Finally, the BOR_temp is SD. 

After baseline, the T4 time point is CR, and ADY=110 exceeds the minimum time limit of SD, which can 
confirm that the BOR_temp is SD. Since T4-T3=27<28, the CR at T4 cannot confirm the CR at T3. 
Finally, the BOR_temp is SD. 

Finally, the BOR at each point in time is as follows: 

 
Table 7. Subject 012 BOR results 

2.2. CONFIRMATION OF RESPONSES AT ADJACENT TIME POINTS AS WELL AS NON-
ADJACENT TIME POINTS 

USUBJID Time_Point ADY RSDY_lag DIFF PARAM PARAMCD RSSTRESC RSEVAL BOR_TEMP1 BOR
012 2 30  Overall Response OVRLRESP CR INVESTIGATOR
012 3 56 30 26 Overall Response OVRLRESP CR INVESTIGATOR
012 4 83 56 27 Overall Response OVRLRESP CR INVESTIGATOR
012 5 110 86 24 Overall Response OVRLRESP CR INVESTIGATOR

RSSTRESC RSEVAL BOR_TEMP1 BOR
CR INVESTIGATOR NE NE
CR INVESTIGATOR SD SD
CR INVESTIGATOR SD SD
CR INVESTIGATOR SD SD

Table 8. Confirmation of responses at adjacent time points as well as non-adjacent time points 
Confirmation of BOR is more complicated when consideration of non-adjacent time point confirmation. 
Rules for reponse at non-adjacent time points are illustrated on the right side of the top. During the 
exercise, there was an unlimited amount of NE and SD between the two PR efficacy, and the first PR was 
considered confirmed. In the case of 3 consecutive CRs, if T2-T1<28, but T3-T1≥28, it is stipulated that 
the 3rd CR can confirm the first CR. PR situation, the same goes for it. 

Column BOR_temp1 in Exercise 2 represents the confirmation of the response at adjacent time points, 
the BOR_temp2 column represents the case where the non-adjacent response is confirmed as CR, and 
the BOR_temp3 column represents the situation where the non-adjacent response is confirmed as PR. 
BOR_temp column will take the best effect of the first 3 columns. 

Subject 006 
The distance between T4 and T2 is 39+31=70>28, and all T4 time point PRs can confirm T2 time point 
PR, and the BOR_temp3 of T4 time point is PR and BOR_temp time point is PR. 

The specific results are as follows: 

 
Table 9. Subject 006 

Subject 007 
The distance between T4 and T1 is 165-38=127>28, and all T4 CRs can confirm the CR at T1, and the 
BOR_temp2 at T4 is CR and BOR_temp is CR. 

The specific results are as follows: 

 
Table 10. Subject 006 

3. SAS PROGRAM FOR BOR CONFIRMATION 
Below is my SAS program for confirming BOR. The program is divided into two parts: one for confirming 
adjacent time point responses and one for confirming non-adjacent time point responses. The non-
adjacent time point confirmation includes both CR and PR. 

USUBJID Time_Point ADY RSDY_lag DIFF PARAM PARAMCD RSSTRESC RSEVAL BOR_TEMP1
BOR_TEMP2

(CR)
BOR_TEMP3

(PR) BOR_TEMP BOR
006 2 30  Overall Response OVRLRESP SD INVESTIGATOR NE NE NE
006 3 61 30 31 Overall Response OVRLRESP PR INVESTIGATOR SD SD SD
006 4 92 61 31 Overall Response OVRLRESP NE INVESTIGATOR SD SD SD
006 5 131 92 39 Overall Response OVRLRESP PR INVESTIGATOR SD PR PR PR

USUBJID Time_Point ADY RSDY_lag DIFF PARAM PARAMCD RSSTRESC RSEVAL BOR_TEMP1
BOR_TEMP2

(CR)
BOR_TEMP3

(PR) BOR_TEMP BOR
007 2 38  Overall Response OVRLRESP CR INVESTIGATOR SD SD SD
007 3 85 38 47 Overall Response OVRLRESP NE INVESTIGATOR SD SD SD
007 4 123 85 38 Overall Response OVRLRESP NE INVESTIGATOR NE NE SD
007 5 165 123 42 Overall Response OVRLRESP CR INVESTIGATOR SD CR CR CR

The confirmation of adjacent time point responses needs to consider all possible combinations, as shown 
in Table 3 of Section 1.5.1. For non-adjacent time point responses, the program mainly focuses on how to 
retain the first CR or PR and how to determine the continuity of the time points. 

The program's output can be compared with manual determinations for verification. 
***1. Get Data from RS; 
proc import datafile = "E:\999_test\Examples_BOR_Confirmation v1.1.xlsx" 
    out = rs dbms = xlsx replace; 
    sheet = "RS"; 
    getnames = yes; 
run; 
 
 
 
***2. Derive BOR_temp; 
 
**2.1 Create format for response; 
proc format; 
    value resp 
        1 = "CR" 
        2 = "PR" 
        3 = "SD" 
        4 = "PD" 
        5 = "NE" 
        . = " " 
    ; 
 
    invalue respn 
        "CR" = 1 
        "PR" = 2 
        "SD" = 3 
        "PD" = 4 
        "NE" = 5 
    ; 
run; 
 
 
**2.2 Variables for analysis; 
data adrs1; 
    set rs; 
    length paramcd $8; 
    paramcd = rstestcd; 
    visitnum = time_point; 
    ady = rsdy; 
    avalc = rsstresc; 
    aval = input (avalc, respn.); 
 
    proc sort; 
        by usubjid ady; 
run; 
 
 
**2.3 Neighboring & Non-neighboring response confirmation; 
data adrs; 
    set adrs1; 
    by usubjid; 

    *Pre RESP; 
    avalc_lag = lag(avalc); 
    aval_lag = lag(aval); 
    ady_lag = lag(ady); 
 
    retain ady_cr ord_cr ady_pr ord_pr born; 
 
    *For first record; 
    if first.usubjid then do; 
 
        *No lag record for first record; 
        call missing(avalc_lag, aval_lag, ady_lag); 
 
        *First bor_temp; 
        if avalc in ("CR" "PR" "SD") then do; 
            if ady>=35 then born_temp1=3; 
            else born_temp1 = 5; 
        end; 
        else if avalc = "PD" then born_temp1 = 4; 
        else if avalc = "NE" then born_temp1= 5; 
 
        *Get bor for first record; 
        born = born_temp1; 
 
        *First CR ady and ord for comfirmation continuity ; 
        if aval = 1 then do; 
            ady_cr = ady; 
            ord_cr = 1; 
        end; 
        else do; 
            ady_cr = .; 
            ord_cr = .; 
        end; 
 
        *First PR ady and ord for comfirmation continuity ; 
        if aval = 2 then do; 
            ady_pr = ady; 
            ord_pr = 1; 
        end; 
        else do; 
            ady_pr = .; 
            ord_pr = .; 
        end; 
    end; 
 
    *For non-first records; 
    else do; 
 
        *For Non-neighboring CR confirmation (ignore NE, CR); 
        if avalc not in ("NE" "CR") then do; 
            ady_cr = .; 
            ord_cr = .; 
        end; 
        else if ady_cr ne . then ord_cr = ord_cr + 1; 
 
        if avalc = "CR"  and ady_cr = . then do; 
            ady_cr = ady; 

            ord_cr = 1; 
        end; 
 
        if avalc = "CR" and ady_cr ne . then do; 
            if ady -ady_cr >=28 and ord_cr >2 then born_temp2 = 1;  
/*ord_cr >2 --> Non-neighboring*/ 
        end; 
 
        *For Non-neighboring  PR confirmation (ignore NE, SD, PR, CR); 
        if avalc not in ("SD" "NE" "PR" "CR") then do; 
            ady_pr = .; 
            ord_pr = .; 
        end; 
        else if ady_pr ne . then ord_pr = ord_pr + 1; 
 
        if avalc = "PR" and ady_pr = . then do; 
            ady_pr = ady; 
            ord_pr = 1; 
        end; 
 
        if avalc in ("CR" "PR") and ady_pr ne . then do; 
            if ady - ady_pr >=28 and ord_pr >2 then born_temp3 = 2; 
/*ord_pr >2 --> Non-neighboring*/ 
        end; 
 
        *avalc_lag="CR"; 
        if avalc_lag="CR" and ady_lag>=35 then do; 
 
            if avalc="CR" then do; 
                if ady-ady_lag>=28 then born_temp1 = 1; 
                else born_temp1 = 3; 
            end; 
 
            else if avalc in ("PR" "SD") then do; 
                born_temp1 = 0; 
                put "War" "ning: need to query data"; 
            end; 
 
            else if avalc in ("PD" "NE" "") then born_temp1 = 3; 
        end; 
 
        else if avalc_lag="CR" and ady_lag<35 then do; 
 
            if avalc="CR" then do; 
                if ady-ady_lag>=28 then born_temp1=1; 
                else if ady-ady_lag<28 and ady>=35 then born_temp1=3; 
                else if ady-ady_lag<28 and ady<35 then born_temp1=5;  
            end; 
 
            else if avalc in ("PR" "SD") then do; 
                born_temp1 = 0; 
                put "War" "ning: need to query data"; 
            end; 
 
            else if avalc = "PD" then born_temp1=4; 
 
            else if avalc in ("NE" "") then born_temp1=5; 

        end; 
 
        *avalc_lag="PR"; 
        if avalc_lag="PR" and ady_lag>=35 then do; 
            if avalc in ("CR" "PR") then do; 
                if ady-ady_lag>=28 then born_temp1 = 2; 
                else born_temp1 = 3; 
            end; 
 
            else if avalc in ("SD" "PD" "NE" "") then born_temp1 = 3; 
        end; 
 
        if avalc_lag="PR" and ady_lag<35 then do; 
            if avalc in ("CR" "PR") then do; 
                if ady-ady_lag>=28 then born_temp1=2; 
                else if ady-ady_lag<28 and ady>=35 then born_temp1=3; 
                else if ady-ady_lag<28 and ady<35 then born_temp1=5;  
            end; 
 
            else if avalc="SD" then do; 
                if ady>=35 then born_temp1=3; 
                else born_temp1 = 5; 
            end; 
 
            else if avalc = "PD" then born_temp1=4; 
 
            else if avalc in ("NE" "") then born_temp1=5; 
        end; 
 
        *avalc_lag="SD"; 
        if avalc_lag="SD" and ady_lag>=35 then born_temp1=3; 
 
        if avalc_lag="SD" and ady_lag<35 then do; 
            if avalc in ("CR" "PR" "SD") then do; 
                if ady>=35 then born_temp1 = 3; 
                else born_temp1 = 5; 
            end; 
 
            else if avalc = "PD" then born_temp1=4; 
 
            else if avalc in ("NE" "") then born_temp1=5; 
        end; 
 
        *avalc_lag="PD"; 
        if avalc_lag="PD" then born_temp1 = 4; 
 
        *avalc_lag="NE"; 
        if avalc_lag = "NE" then do; 
            if avalc in ("CR" "PR" "SD") then do; 
                if ady>=35 then born_temp1 = 3; 
                else born_temp1 = 5; 
            end; 
 
            else if avalc = "PD" then born_temp1=4; 
 
            else if avalc in ("NE" "") then born_temp1=5; 
        end; 

    end; 
 
    *born for or non-first records; 
    if born > min(born_temp1, born_temp2, born_temp3) then born = 
min(born_temp1, born_temp2, born_temp3); 
 
    bor_temp1 = put(born_temp1, resp.); 
    bor_temp2 = put(born_temp2, resp.); 
    bor_temp3 = put(born_temp3, resp.); 
    bor = put(born, resp.); 
 
    keep usubjid paramcd visitnum born_temp1 bor_temp1 born_temp2 bor_temp2 
born_temp3 bor_temp3 born bor aval avalc ady ord_cr ord_pr; 
run; 

CONCLUSION 
This article provides a comprehensive guide to confirming the BOR in oncology clinical trials using the 
RECIST 1.1 criteria. By thoroughly detailing the procedures for both adjacent and non-adjacent time point 
response confirmations, it aims to equip researchers with the knowledge to accurately confirm BOR.  

The practical exercises and SAS programming examples included are intended to enhance readers' 
understanding and proficiency in applying these criteria. By mastering these methods, researchers can 
ensure that the confirmation of BOR is robust and reliable, ultimately contributing to more precise and 
effective clinical trial outcomes. 

